
Optimization Methods
Draft of August 26, 2005

V.
Solving Integer
Linear Programs

Robert Fourer

Department of Industrial Engineering and Management Sciences
Northwestern University
Evanston, Illinois 60208-3119, U.S.A.

(847) 491-3151

4er@iems.northwestern.edu
http://www.iems.northwestern.edu/˜4er/

Copyright c 1989–2004 Robert Fourer



B–126 Optimization Methods — x15.2



Draft of August 26, 2005 B–127

16. Solving by Branch-and-Bound

Mixed integer programs are fundamentally harder to solve than ordinary lin-
ear programs, because some of the nice properties of LPs do not carry over to
the integer case. Given a feasible solution to a mixed integer program, there
is no simple criterion (like an LP’s reduced costs) for either finding a direction
toward a better solution or determining that the current solution is optimal.
As a result, there is no general algorithm for mixed integer programs that iter-
ates directly to an optimum by stepping through successively better and better
solutions.

The only way to be sure that you have an optimal mixed-integer solution is
to either try or rule out every possibility for the integer variables. The branch-
and-bound approach is a way of carrying out this enumeration, via the solutions
to a series of linear programs, so that the great majority of infeasible and sub-
optimal possibilities are in fact ruled out. Branch-and-bound has the further
advantage of providing progressively tighter lower and upper bounds on the
objective value, so that it may be stopped early with a provably good solution.
Even so, this approach is not reliable to the same degree as linear programming
methods. It may attempt to search millions or billions of possibilities, even for
a problem of only modest size. Sometimes its performance can be greatly im-
proved by altering the problem formulation, or by changing key details of the
search procedure.

In this chapter we illustrate the fundamental ideas of branch-and-bound, by
giving a complete example of its application to a small zero-one mixed integer
program. A summary of the steps in general terms is given at the end of the
chapter.

Practical solvers for integer programming are based upon this approach, but
incorporate many modifications and extensions to speed the search. Some of
these refinements will be discussed in a subsequent chapter to be written.

16.1 A sample problem

To introduce branch-and-bound, we use a very simple transportation model
with capacities, variable costs, and fixed costs on the links from origins to
destinations. The key to our formulation, shown in Figure 16–1, is the defin-
ition of zero-one variables Use[i,j] that are 1 if a positive amount is shipped
from origin i to destination j, and 0 otherwise, along with continuous variables
Trans[i,j] that denote the amounts shipped from i to j. The expression

fcost[i,j] * Use[i,j] + vcost[i,j] * Trans[i,j]

in the objective thus represents the total cost associated with shipping from i
to j. In addition to the standard supply and demand constraints for a trans-
portation model, the constraint

Trans[i,j] <= capacity[i,j] * Use[i,j]

implies that at most capacity[i,j] is shipped when Use[i,j] is 1, and that

set ORIG;
set DEST;

param supply {ORIG} >= 0;
param demand {DEST} >= 0;
param capacity {ORIG,DEST} >= 0;

param vcost {ORIG,DEST} >= 0;
param fcost {ORIG,DEST} >= 0;

var Use {ORIG,DEST} binary;
var Trans {ORIG,DEST} >= 0;

minimize Total Cost:
sum {i in ORIG, j in DEST}

(fcost[i,j] * Use[i,j] + vcost[i,j] * Trans[i,j]);

subj to Supply {i in ORIG}: sum {j in DEST} Trans[i,j]<=supply[i];

subj to Demand {j in DEST}: sum {i in ORIG} Trans[i,j]>=demand[j];

subj to Cqp Use {i in ORIG, j in DEST}:
Trans[i,j] <= capacity[i,j] * Use[i,j];

Figure 16–1. Mixed-integer model for branch-and-bound example (fixcost.mod).

set ORIG := A B C;
set DEST := Y Z;

param supply := A 175 B 175 C 175 ;

param demand := Y 200 Z 200 ;

param: capacity fcost vcost :=
A Y 150 200 10
A Z 75 300 10
B Y 125 400 20
B Z 150 400 30
C Y 160 250 15
C Z 100 500 20 ;

Figure 16–2. Sample data for the branch-and-bound example (fixcost.dat).

zero is shipped when Use[i,j] is 0.
To keep our example manageable, we consider an instance with three origins

and two destinations, given by the data in Figure 16–2.
There are only 6 zero-one variables in this problem, and hence only 26 �

64 different ways that they can be assigned the values zero and one. Thus,
for purposes of comparison, we can solve this problem by fixing the Use[i,j]
variables to each of the 64 different combinations, and then using a regular LP
solver to find the best setting of the Trans[i,j] variables. It turns out that for

54 of the combinations, there is no feasible solution. The best values for the
remaining 10 are:

From: A A B B C C
To: Y Z Y Z Y Z Total Cost

0 1 1 1 1 0 9050.0000
0 1 1 1 1 1 8975.0000
1 0 0 1 1 1 8600.0000
1 0 1 1 0 1 9000.0000
1 0 1 1 1 1 9000.0000
1 1 0 1 1 0 8150.0000
1 1 0 1 1 1 7900.0000
1 1 1 1 0 1 8300.0000
1 1 1 1 1 0 8550.0000
1 1 1 1 1 1 8175.0000

Thus the minimum cost is 7900, achieved by using all routes except B to Y.
The optimal shipment amounts are given by fixing Use["B","Y"] to zero and
all other Use[i,j] to 1, and solving the resulting linear program for all of the
Trans[i,j] variables:

ampl: model fixtrans.mod;
ampl: data fixtrans.dat;

ampl: fix Use["A","Y"] := 1;
ampl: fix Use["A","Z"] := 1;
ampl: fix Use["B","Y"] := 0;
ampl: fix Use["B","Z"] := 1;
ampl: fix Use["C","Y"] := 1;
ampl: fix Use["C","Z"] := 1;

ampl: solve;
MINOS 5.5: optimal solution found.
4 iterations, objective 7900

ampl: option display 1col 0;
ampl: option display eps .000001;

ampl: display Trans;

Trans [*,*]
: Y Z :=
A 100 75
B 0 50
C 100 75
;

We can use this solution to check our results later. This overall approach is
not practical, however, for any but the smallest problems. If the number of
destinations is increased from 2 to 10, for example, then there are 30 zero-one
variables, giving rise to 230 � 1;073;741;824 different solutions. Each addi-
tional zero-one variable further doubles the number of different solutions to be
tried, and the situation is even worse for integer variables that have more than
two feasible values. Mixed integer programs of practical interest often have
hundreds or thousands of integer variables.

16.2 An example

The idea of the branch-and-bound approach in the simplest case — where
all of the integer variables are zero-one — is to fix the variables only one at
a time, re-applying an LP solver after each fixing. The values of some integer
variables that are not yet fixed may take fractional values, but these fractional
solutions are able to provide valuable information that can be used to rule out
many combinations of integer values as candidates for the optimum. As a result,
the number of different linear programs to be solved may be far less than the
number of different ways in which the integer variables can be assigned values.

This section introduces branch-and-bound by concentrating on the funda-
mental ideas for determining bounds and for reducing the number of LPs solved.
It also defines the fundamental terminology of the “search tree” of solutions. We
defer to the next section a discussion of refinements and extensions to speed
the procedure and to generalize it beyond zero-one variables.

The branch-and-bound process begins by solving the relaxation of the in-
teger program, in which all of the integrality restrictions are dropped. For the
case of zero-one variables, the relaxation imposes a lower bound of zero and an
upper bound of one, but also allows any fractional values in between. For our
example, we can set AMPL’s relax integrality option to 1 to suppress the
communication of all integer restrictions to the solver:

ampl: unfix Use;
ampl: option presolve 0;
ampl: option relax integrality 1;

ampl: solve;
MINOS 5.5: optimal solution found.
11 iterations, objective 7322.1875

ampl: display Use;

Use [*,*]
: Y Z :=
A 0.666667 1
B 0.2 0.166667
C 0.46875 1
;

ampl: display Trans;

Trans [*,*]
: Y Z :=
A 100 75
B 25 25
C 75 100
;

Since the relaxation only enlarges the region of feasible solutions, its minimum
is at least as low as the original problem’s minimum. Thus the objective value
that it finds, 7322.1875, is a lower bound on the original problem’s objective.
This is our example’s first instance of the “bounding” aspect of branch-and-
bound.

Because our solution has four variables that lie at fractional values between
0 and 1, it cannot be optimal for the integer program. We thus proceed by
considering what happens when one of the fractional variables is fixed at 0 or
at 1. We can pick any one fractional variable for this purpose; the branch-and-
bound procedure will eventually succeed regardless of our choice. Some choice
criteria may lead to a much more efficient search than others, but we’ll leave
that possibility for later discussion.

Suppose that we pick the fractional variable Use["B","Z"] to be fixed. Then
we can fix this variable to 0 or 1, and re-solve for the other five variables. We’ll
refer to the resulting restricted LP relaxations as branch-and-bound subprob-
lems. In the subproblem with Use["B","Z"] fixed to 0,

ampl: fix Use["B","Z"] := 0;

ampl: solve;
MINOS 5.5: infeasible problem.
0 iterations

we find that there is no feasible solution. This tells us that there cannot pos-
sibly be a feasible solution, and in particular an integer feasible solution, when
Use["B","Z"] is 0. In the subproblem where Use["B","Z"] is instead fixed
to 1,

ampl: fix Use["B","Z"] := 1;

ampl: solve;
MINOS 5.5: optimal solution found.
2 iterations, objective 7614.583333

ampl: display Use;

Use [*,*]
: Y Z :=
A 0.666667 1
B 0 1
C 0.625 0.75
;

we get a solution having a higher objective value, but still some fractional vari-
ables. Because fixing a variable is a restriction of the problem, it will always tend
to increase the objective of a minimization. The effect on the optimal solution
is not so easy to predict, however. Notice that although Use["B","Z"] is much
closer to 0 than to 1 in the relaxation, we have had to fix it to 1 to get a feasible
subproblem. In the solution to this subproblem, the variable Use["B","Y"],
which was previously fractional, now has an integer value, while Use["C","Z"],
which was previously integral, is now at a fraction value. In general, there is no
easy way of telling how fixing one variable will affect the optimal values of the
remaining integer variables that have not been fixed. The only way to find out
is to actually fix the variable and call the solver to re-optimize.

What have we gained by invoking the solver two more times? First, we have
found that there is no feasible solution with Use["B","Z"] at 0. As a result,
we need not do any further work to consider any of the 32 different integer

solutions that have 0 as the value of Use["B","Z"]. Second, we have found that
any solution with Use["B","Z"] at 1 has an objective value of 7614.583333 or
more. Taken together, these outcomes imply that 7614.583333 is a new, higher
lower bound on the objective value for integer solutions.

To keep track of the branch-and-bound process, we’ll find it convenient to
summarize our progress in the form of a search tree like this:

7322

7614infeas

Use["B","Z"]
0 1

The circles are the nodes of the tree, which correspond to subproblems — LP
relaxations with certain variables fixed. The number in each circle is the sub-
problem’s objective value. At the top is the root node corresponding to the first
problem that we sent to the solver, with all of the integrality conditions relaxed.
From the root there are two branches, corresponding (as marked) to the fixing
of Use["B","Z"] at zero and at one. Each branch leads to a child node that
corresponds to a subproblem that is solved after Use["B","Z"] is fixed.

The left-hand child node of the root, where the LP relaxation is infeasible,
need not be explored further; we say that it has been fathomed. The right-
hand child requires more attention, however, because there is still the possibility
of various feasible integer solutions with Use["B","Z"] fixed to 1. We thus
proceed to perform another step of branch-and-bound at this child node. We
select one of the three variables that is fractional in the subproblem solution at
this node — say, Use["C","Z"] — and solve subproblems in which this second
variable is also fixed at 0,

ampl: fix Use["C","Z"] := 0;

ampl: solve;
MINOS 5.5: optimal solution found.
2 iterations, objective 7989.583333

ampl: display Use;

Use [*,*]
: Y Z :=
A 0.666667 1
B 0 1
C 0.625 0
;

or at 1:

ampl: fix Use["C","Z"] := 1;

ampl: solve;
MINOS 5.5: optimal solution found.
3 iterations, objective 7655.520833

ampl: display Use;

Use [*,*]
: Y Z :=
A 0.666667 1
B 0.2 1
C 0.46875 1
;

In this case, both of the subproblems are feasible, and the search tree has been
extended as follows:

7655 7989

1 0
Use["C","Z"]

7322

7614infeas

Use["B","Z"]
0 1

The results tell us that, with Use["B","Z"] already fixed at 1, any solution with
also Use["C","Z"] at 1 must have an objective value of at least 7655.520833,
while any solution with Use["C","Z"] at 0 must have an objective value of
at least 7989.583333. Since these are the only possibilities, we can conclude
any integer solution at all must have an objective value at least as great as the
smaller of these two — 7655.520833 — and hence the optimal integer solution
much have an objective value at least this great. Thus once again our lower
bound on the optimal value has been increased.

We now have two active nodes and can proceed by doing another branching
step at either one. Suppose that we branch from the node at the left. Then the
next variable to fix to 0 or 1 would naturally be one of those that is fractional
at that node’s subproblem; there are still three possibilities, Use["A","Y"],
Use["B","Y"], and Use["C","Y"]. If we take, for example, Use["C","Y"], we
find that there are again feasible solutions in which it is fixed at 0,

ampl: fix Use["C","Y"] := 0;

ampl: solve;
MINOS 5.5: optimal solution found.
0 iterations, objective 8153.333333

ampl: display Use;

Use [*,*]
: Y Z :=
A 0.666667 1
B 0.8 1
C 0 1
;

and at 1:

ampl: fix Use["C","Y"] := 1;

ampl: solve;
MINOS 5.5: optimal solution found.
1 iterations, objective 7788.333333

ampl: display Use;

Use [*,*]
: Y Z :=
A 0.666667 1
B 0.2 1
C 1 1
;

Two more branches are added to the search tree, increasing the number of active
nodes to three:

8153 7788

0 1
Use["C","Y"]

7655 7989

1 0
Use["C","Z"]

7322

7614infeas

Use["B","Z"]
0 1

Any feasible solution to our original mixed integer problem must be the result
of fixing further variables under one of the active nodes, and hence must have
an optimal objective value greater than or equal to the optimal value at one
of the active nodes. Thus we can conclude that a lower bound on the optimal
solution to the mixed integer program is given by the minimum of the objective
values at the active nodes — in this case, by 7788.333333. This same reasoning
applies throughout the branch-and-bound procedure. Although the resulting

lower bounds do not increase at every branching step, they do tend to become
tighter as the search proceeds.

The solutions at both of the two new child nodes have only two fractional
variables, Use["A","Y"] and Use["B","Y"]. Suppose that we choose to branch
first on the left-hand node and to fix Use["B","Y"] there. We find that the only
feasible possibility is to fix it to 1:

ampl: fix Use["C","Y"] := 0;
ampl: fix Use["B","Y"] := 0;

ampl: solve;
MINOS 5.5: infeasible problem.
2 iterations

ampl: fix Use["B","Y"] := 1;

ampl: solve;
MINOS 5.5: optimal solution found.
2 iterations, objective 8233.333333

ampl: display Use;

Use [*,*]
: Y Z :=
A 0.666667 1
B 1 1
C 0 1
;

Then if we branch on the only remaining fractional variable, Use["A","Y"], we
find again that the only feasible possibility is to fix it to 1:

ampl: fix Use["A","Y"] := 0;

ampl: solve;
MINOS 5.5: infeasible problem.
0 iterations

ampl: fix Use["A","Y"] := 1;

ampl: solve;
MINOS 5.5: optimal solution found.
2 iterations, objective 8300

ampl: display Use;

Use [*,*]
: Y Z :=
A 1 1
B 1 1
C 0 1
;

At this point we have reached our first integer solution. The search tree looks
like this, with the node giving the integer solution indicated by a double circle:

8233

1 0
Use["B","Y"]

infeas

1 0
Use["A","Y"]

infeas

8300

8153 7788

0 1
Use["C","Y"]

7655 7989

1 0
Use["C","Z"]

7322

7614infeas

Use["B","Z"]
0 1

We had to fix only five of the integer variables to get to this solution. The
sixth, Use["A","Z"], happens to take an integer value at the optimum for the
subproblem, without being fixed explicitly.

Because this most recent solution is feasible for our mixed integer program,
it achieves an objective value that is greater than or equal to the optimal value.
That is, it gives an upper bound of 8300 for the optimum. Moreover, because the
solution to the subproblem shows that the best setting for the unfixed variable
Use["A","Z"] is 1, there is no need to consider fixing this variable to 0. We
see here a second situation in which a node may be fathomed: when all unfixed
variables take integer values in the solution to the subproblem.

We now have both a lower bound and an upper bound for the mixed integer
program. Thus we can say that its optimal value lies somewhere between them
— that is, between 7788.333333 and 8300. To say the same thing another way,
the best integer solution found so far can’t be more than about 6.6% worse than
the optimum in terms of its objective value.

In fact we may already have found the minimum-cost solution to the mixed
integer program. We can’t be sure whether we have found it, however, until the
branch-and-bound procedure has finished exploring the two remaining active
nodes of the search tree. There is no simple algebraic criterion (like nonneg-
ativity of reduced costs for LPs) that can be applied to verify optimality of an
integer programming solution in all cases. In some situations, the effort in-
volved in finding an optimal solution can be less than the effort of proving that
it is optimal. In such cases it often makes sense to stop once the lower bound is
acceptably close to the upper bound; we’ll come back to this in the next section.

To continue the branch-and-bound procedure, we again select an active node
and a variable to branch on at that node. Suppose that we select the node
where Use["B","Z"], Use["C","Z"], and Use["C","Y"] are all fixed to 1 —

that is, the node labeled with the subproblem objective of 7788. We previously
observed that Use["A","Y"] and Use["B","Y"] were the only fractional vari-
ables left at that point, so suppose we branch on Use["A","Y"]. Here are the
solutions that we get when fixing this variable to 0 and to 1:

ampl: fix Use["C","Y"] := 1;
ampl: unfix Use["B","Y"];

ampl: fix Use["A","Y"] := 0;

ampl: solve;
MINOS 5.5: optimal solution found.
4 iterations, objective 8975

ampl: fix Use["A","Y"] := 1;

ampl: solve;
MINOS 5.5: optimal solution found.
1 iterations, objective 7855

ampl: display Use;

Use [*,*]
: Y Z :=
A 1 1
B 0.2 1
C 1 1
;

When Use["A","Y"] is fixed to 0, the minimum objective value for the sub-
problem jumps to 8975. Since we have already found an integer solution that
costs only 8300, however, there is nothing to be gained by further considering
the solution to this subproblem, or to the subproblems at any children it might
have. They are all going to have objective values of 8975 or greater, which are
of no interest when we already know how to achieve an objective of 8300.

Thus we can mark the node where Use["A","Y"] is fixed to 0 as fathomed,
and need not consider it further. This is an example of the third (and final) kind
of situation in which fathoming is possible: where the subproblem objective is
greater than or equal to the objective value of the best integer-feasible solution
found so far — or equivalently, where the subproblem objective is greater than
or equal to the current upper bound.

The sole active child node just created has only one remaining fractional
variable, Use["B","Y"], so for the next branch-and-bound step we can consider
fixing that variable to 0 or 1:

ampl: fix Use["B","Y"] := 0;

ampl: solve;
MINOS 5.5: optimal solution found.
3 iterations, objective 7900

ampl: display Use;

Use [*,*]
: Y Z :=
A 1 1
B 0 1
C 1 1
;

ampl: fix Use["B","Y"] := 1;

ampl: solve;
MINOS 5.5: optimal solution found.
1 iterations, objective 8175

Fixing Use["B","Y"] to 0 brings us to another all-integer subproblem solution.
Its objective value, 7900, improves on the best we previously found, and so
becomes our new upper bound.

The search tree at this point is as follows:

8975 7855

0 1

7900 8175

Use["A","Y"]

Use["B","Y"]

0 1

8233

1 0
Use["B","Y"]

infeas

1 0
Use["A","Y"]

infeas

8300

8153 7788

0 1
Use["C","Y"]

7655 7989

1 0
Use["C","Z"]

7322

7614infeas

Use["B","Z"]
0 1

There are still two active nodes left, but at both of them the subproblem ob-
jective is higher than our new upper bound of 7900. Thus both can be marked
as fathomed immediately. Since that leaves us with no nodes remaining to be
searched, we can finally conclude that 7900 is the minimum cost achievable.

The completed search tree contains only 15 nodes. Thus we have found the
optimal solution, and have confirmed that it is optimal, by solving only 15 linear
programs, rather than the 64 that would be necessary if we simply tried setting
the Use variables to all of the possible combinations of 0 and 1. For considerably

larger problems of practical interest, the set of possible combinations is far too
large to consider, and yet the branch-and-bound approach often succeeds in
establishing an optimum after solving a manageable number of LPs. Its success
does often rely on a number of enhancements, some of which are discussed in
more advanced sections.

16.3 A general statement

The simplified version of branch-and-bound used in our example is easily
summarized in more general terms. The following description assumes a mini-
mization, but only a few minor changes are necessary to adapt it to a maximiza-
tion.

Initially, the search tree consists of one active node, the root node. The
value of this node is the objective value of the continuous relaxation of the
mixed integer program of interest. The initial lower bound is the value of the
root node, and the initial upper bound is 1 (or the objective value of a feasible
solution that happens to be known or easily computed).

If the solution at the root node is integer-feasible (meaning that all integer
variables are given integer values), then the solution is optimal for the mixed
integer program and there is no further work to be done. Otherwise, some of the
integer variables have fractional values, and the branch-and-bound procedure
consists of the following steps:

1. Select an active node.

2. Select a fractional-valued variable from the solution at that node.

3. Form two child nodes, by fixing the variable to 0 and to 1.

4. Solve the subproblems at the child nodes.

5. Check whether either subproblem can be fathomed.

6. Update the lower and upper bounds, and the set of active nodes.

These steps are repeated as long as any active nodes remain. When the set
of active nodes is empty, the best integer-feasible solution found so far is the
optimal solution to the mixed-integer program.

In step 5, there are three possibilities for fathoming each node:

5a. The subproblem has no feasible solution.

5b. The subproblem’s optimal value is � the upper bound.

5c. The subproblem’s optimal solution is integer-feasible.

In 5a and 5b, the node and its children cannot possibly give an optimal solution
to the mixed-integer program. If 5c holds and the node has not already been
fathomed under 5b, then the subproblem’s solution must give a better objective
value than any other integer-feasible solution found so far; its solution is saved
in case it does turn out to be the best.

In step 6, the selected parent node is made inactive, and any child node not
fathomed is made active. The bounds are adjusted to maintain the following
properties:

6a. The lower bound equals the minimum of the subproblem objec-
tive values at the active nodes.

6b. The upper bound equals the objective value of the best integer-
feasible solution found so far.

Thus the lower bound only needs updating when the selected node’s subprob-
lem objective value is equal to the lower bound. The upper bound changes only
when an improved integer-feasible solution is found — case 5c above — in which
case the upper bound is reset to the improved solution’s objective value.




